Portfolio Optimization

Xiangyi Meng

Utility and Indifference Curves

What's your choice?

a) a riskless payment of $10;

b) a 50/50 chance of $20 or $0.

- **Investors are typically thought of as being risk averse.**
- **A trade-off exists between return and risk.**

Indifference curve?

The investors have no preference for one combination over a different combination on the same curve.

In[1]:=
```
ClearAll["Global`*"];

U[μ_, σ_] = μ - σ^2;

g = ContourPlot[U[μ, σ], {σ, 0, 0.02}, {μ, 0, 0.001},
  PlotLegends → Automatic,
  ImageSize → Large,
  FrameLabel → {"standard deviation", "mean return"},
  BaseStyle -> {FontSize → 20}]
```

Out[3]=



Correlation: Two Risky Assets

In[4]:=
```
μa = 0.0008; σa = 0.015; μb = 0.0006; σb = 0.01;
(* Allocation on stock A(B) is p (1-p) *)
(* ρab is the correlation between stocks A and B*)
Manipulate[
 Show[g,
  ParametricPlot[{√(p^2 σa^2 + (1 - p)^2 σb^2 + 2 p (1 - p) ρab σa σb), p μa + (1 - p) μb},
   {p, 0, 1},
   ImageSize → Large,
   BaseStyle -> {FontSize → 20}]
 ],
 {ρab, -1, 1}]
```

Out[5]=



Negative correlation means canceling out the risk!

Optimal Asset Allocation

GLD SPDR Gold Trust (ETF)
USO United States Oil Fund LP (ETF)
UUP PowerShares DB US Dollar Index Bullish Fund
SPY SPDR S&P 500 ETF Trust
ONEQ Fidelity NASDAQ Comp. Index Trk Stk(ETF)
TLO SPDR Lehman Long Term Treasury (ETF)
BIL SPDR Lehman 1-3 Month T Bill (ETF)

```
strings = {"GLD", "USO", "UUP", "SPY", "ONEQ", "TLO", "BIL"};
prices = FinancialData[#, {{2010}, {2016}}, "Value"] & /@ strings;
ListLogPlot[prices,
 PlotLegends → SwatchLegend[strings],
 ImageSize → Large,
 AxesLabel → {"time", "price"},
 BaseStyle -> {FontSize → 20}]
```



Optimal Asset Allocation

```
returns = Differences /@ (Log /@ prices);
means = Mean[Transpose[returns]];
cov = Covariance[Transpose[returns]];

means // MatrixForm
cov // MatrixForm
```

$$\begin{pmatrix} -0.0000523499 \\ -0.000859981 \\ 0.000074575 \\ 0.000469903 \\ 0.000548642 \\ 0.000301172 \\ -2.43271\times10^{-6} \end{pmatrix}$$

$$\begin{pmatrix} 0.000120521 & 0.0000495686 & -0.0000177737 & 6.19459\times10^{-6} & 5.63544\times10^{-6} & 0.0000124415 & -8.19016 \\ 0.0000495686 & 0.000339106 & -0.0000283937 & 0.0000866954 & 0.000085965 & -0.0000494165 & -6.84274 \\ -0.0000177737 & -0.0000283937 & 0.0000249935 & -0.0000148187 & -0.0000135036 & 4.00758\times10^{-6} & -5.50696 \\ 6.19459\times10^{-6} & 0.0000866954 & -0.0000148187 & 0.0000992948 & 0.000103046 & -0.0000433514 & -8.99773 \\ 5.63544\times10^{-6} & 0.000085965 & -0.0000135036 & 0.000103046 & 0.000121924 & -0.0000442867 & -9.25329 \\ 0.0000124415 & -0.0000494165 & 4.00758\times10^{-6} & -0.0000433514 & -0.0000442867 & 0.0000694729 & 6.9201\times \\ -8.19016\times10^{-8} & -6.84274\times10^{-8} & -5.50696\times10^{-9} & -8.99773\times10^{-8} & -9.25329\times10^{-8} & 6.9201\times10^{-8} & 2.63309\times \end{pmatrix}$$

```
vars = Map[Subscript[p, #] &, strings];
U[μ_, σ2_] = μ - σ2;
FindMaximum[{
  U[vars.means, vars.cov.vars], Total[vars] == 1 && Apply[And, Thread[Greater[vars, 0]]]}, vars]
allocations = vars /. %[[2]];
```

```
{0.000432671, {p_GLD → 0.000231757, p_USO → 0.0000676044, p_UUP → 0.000405103,
  p_SPY → 0.00214179, p_ONEQ → 0.84523, p_TLO → 0.151652, p_BIL → 0.000270906}}
```

```
ContourPlot[U[μ, σ^2], {σ, 0, 0.02}, {μ, 0, 0.001},
 Epilog → {PointSize[0.04], Point[{Sqrt[allocations.cov.allocations], allocations.means}]},
 PlotLegends → Automatic,
 ImageSize → Large,
 FrameLabel → {"standard deviation", "mean return"},
 BaseStyle -> {FontSize → 20}]
```